DECUS

PROGRAM LIBRARY

DECUS NO.	8-566
TITLE	PARTL
AUTHOR	Andres T. Siy
COMPANY	Capitol Institute of Technology Kensington, Maryland
DATE	July 14, 1972
SOURCE LANGUAGE	8K FORTRAN

Siy

SUB 8K-# Jan. 22, 1972

(1) NAME: PARTL

Purpose: To evaluate the partial fraction expansion o. a rational function N(s)/D(s), that has real coefficients and D(s) are written in linear of quadratic factors.

(2) CALLING SEQUENCE:

CALL PARTL(N,IPOLE,SP,P,Q,RI,AI,BI)

where:

Input Data:

```
N      = deg D(s)
IPOLE = 1, means linear factor in D(s)
      = 0, means quadratic  factor in D(s)
A(I), I=1,2, ..., N  coeff. of N(s)
B(I),I =1,2, ..., N+1  coeff. of D(s)
SP    = (see (4), or (7), or (8) )
P     = ( "   "    "  "    "  "  )
Q     = ( "   "    "  "    "  "  )
```

Return Data:

```
RI    = ( "   "    "  "    "  "  )
AI    = ( "   "    "  "    "  "  )
BI    = ("    "    "  "    "  "  )
```

(3) ERROR RETURN:

-No build in error check, but if computer overflow /underflow the following message will be printed on the Teletype printer,

"FPNT" ERROR AT LOC 03152

(4) SPECIAL CONSIDERATION:

User should make sure that the following strict requirements are followed:

1. degD(s) = N, at least 2
2. degN(s), is at most N-1
3. D(s) must be factored into linear and/ quadratic factors of the form;

$$s + sp, \quad s^2 + ps + q$$

where sp, p and q are real nos.

(5) <u>SUBPROGRAM CALLED:</u>

-none-

(6) <u>STORAGE REQUIRED:</u>

pages (octal)

(7) <u>ALGORITHM & REFERENCES:</u>

<u>Given</u>: $G(s) = N(s)/D(s) = \dfrac{a_1 s^{n-1} + \dots + a_n}{e_1 s^n + e_2 s^{n-1} + \dots + e_{n+1}}$

where $D(s) = \prod_{i=1}^{m} (s+l_i) \prod_{j=1}^{n-m} (s^2 + pj\, s + q_j)$

for some reals l_i, p_j, q_j
and integer (pos) m

<u>Problem</u>: Find

$$G(s) = \sum_{i=1}^{m} \frac{K_i}{s + l_i} + \sum_{j=1}^{n-m} \frac{A_j s + B_j}{s^2 + p_j s + q_j}$$

<u>Solution</u>:

For each i, let l_i = sp, and for each j let p_j, q_j be p and q.
Assume the restriction in (4) hold, then

Algorithms

(<u>CASE I</u>): For each i = 1,2, ..., m

$D(s) = e_1 s^n + \dots + e_{n+1}$
$= (s + l_i)(b_1 s^{n-1} + \dots + b_n)$

step1 ; Generate sequence of nos. $b_1, b_2, \dots, b_n$ and store away, from

$b_1 = e_1$
....
$b_k = e_k - l_i b_{k-1}$ for k=2,3, ..., n

step2 : Construct tw sequences, $c_1, c_2, \dots, c_n$ and $d_1, d_2, \dots, d_n$ and store away, from

$c_1 = a_1$ $\qquad$ $d_1 = b_1$

....

$c_k = -c_{k-1}l_i + a_k$ $\qquad$ $d_k = -d_{k-1}l_i + b_k$

for $k = 2, 3, \ldots, n$

step3 : Then calculate and print,

$K_i = c_n/d_n$

step4 : If $i = m$, go to Case II, otherwise repeat step 1 thru step 4.

(CASE II): For each $j = 1, 2, \ldots, n-m$

$D(s) = e_1 s^n + \ldots + e_{n+1}$

$= (s^2 + p_j s + q_j)(b_1 + \ldots + b_{n-1})$

step1 : Generate seq. $b_1, b_2, \ldots, b_{n-1}$ and store away from

$b_1 = e_1$, $b_2 = e_2 - p_j b1$

....

$b_k = e_k - p_j b_{k-1} - q_j b_{k-2}$ for $k = 3, 4, \ldots, n-1$

set $b_n = 0$ (Important !!)

step2 : Construct 2 sequences $c_1, c_2, \ldots, c_n$ and $d_1, d_2, \ldots, d_n$ and store away, from

$c_1 = a_1$ $\qquad$ $d_1 = b_1$

$c_2 = -c_1 p_j + a_2$ $\qquad$ $d_2 = -d_1 p_j + b_2$

....

$c_k = -c_{k-1} p_j - c_{k-2} q_j + a_k$

$d_k = -d_{k-1} p_j - d_{k-2} q_j + b_k$

for $k = 3, 4, \ldots, n$

step3 : The evaluate, and print

$$A_j = \frac{d_{n-1} c_n - d_{n-1} c_{n-1}}{d_n d_{n-2} - d_{n-1}^2}$$

$$B_j = \frac{d_n c_{n-1} - d_{n-1} cn}{d_n d_{n-2} - d_{n-1}^2}$$

step4 : If j = n-m, FINISHED. Otherwise repeat step 1 thru step 4.

References:

B.O. Watkins, "A Partial Fraction Algorithm," IEEE Trans on Automatic Control vol. Oct. 1971, pp. 489-491.

(8) LISTING:
(See attach)

(9) SAMPLE:
(See attach)

LISTING

```
C ****  PARTIAL FRACTION EXPANSION
C ****  N(S)/D(S)  = ( A(1)*S**(N-1) + ... + A(N) )/
C                    (E(1)*S**N + ......... + E(N+1) )
C       N IS ATMOST 10
C       JAN. 22, 1792
C
C ** REQUIREMENTS:
C       1.  DEG(D(S)) = N,  AT LEAST 2
C       2.  DEG(N(S)) = ATMOST (N-1)
C       3.  D(S) MUST HAVE SIMPLE POLES
C       4.  FACTORS OF D(S)  MUST BE OF THE FORM:
C               S + SP,  S**2 + P*S + Q
C               WHERE:  SP, P & Q ARE REAL NOS.
C       IPOLE = 1, SIMPLE POLES
C             = 2, QUADRATIC FACTORS IN D(S)
        SUBROUTINE PARTL(N,IPOLE,SP,P,Q,RI,AI,BI)
        COMMON A,E
        DIMENSION A(10),E(11),B(10),C(11),D(11)
        IF(IPOLE-1) 10,10,20
10      B(1)=E(1)
        DO 30 K=2,N
30      B(K)=E(K)-SP*B(K-1)
        C(1)=A(1)
        D(1)=B(1)
        DO 40 K=2,N
        D(K)= -D(K-1)*SP + B(K)
40      C(K)= -C(K-1)*SP + A(K)
        RI=C(N)/D(N)
        RETURN
20      B(1)=E(1)
        B(2)=E(2)-P*B(1)
        B(N)=0.
        N1=N-1
        DO 50 K=3,N1
50      B(K)=E(K)-P*B(K-1)-Q*B(K-2)
        C(1)=A(1)
        C(2)=-C(1)*P+A(2)
        D(1)=B(1)
        D(2)=-D(1)*P+B(2)
        DO 60 K=3,N
        D(K)=-D(K-1)*P-D(K-2)*Q+B(K)
60      C(K)=-C(K-1)*P-C(K-2)*Q+A(K)
        SS=D(N)*D(N-2)-D(N-1)**2
        AI=(D(N-2)*C(N)-D(N-1)*C(N-1))/SS
        BI=(D(N)*C(N-1)-D(N-1)*C(N))/SS
        RETURN
        END
```

SAMPLE

```
C ****  SAMPLE # 1
C       CALL  PARTL
C
C       N= DEGREE OF D(S)  ATMOST 10
C       NSP = # SIMPLE POLES
C       NQD = # QUADRATIC FACTORS IN D(S)
C
        COMMON A,B
        DIMENSION A(10),B(11)
        READ(2,10) N,NSP,NQD
10      FORMAT(3I5)
        DO 5 I=1,N
5       READ(2,25) A(I)
        N1=N+1
        DO 6 I=1,N1
6       READ(2,25) B(I)
        IPOLE =1
        I=1
15      IF (I-NSP) 30,30,40
30      READ(2,25) SP
25      FORMAT(E12.5)
C
        CALL PARTL(N,IPOLE,SP,P,Q,RI,AI,BI)
        WRITE(2,25) RI
        I=I+1
        GO TO 15
40      IPOLE =2
        I=1
45      IF (I-NQD) 50,50, 60
50      READ(2,55) P, Q
55      FORMAT(2E12.5)
C
        CALL PARTL(N,IPOLE,SP,P,Q,RI,AI,BI)
        WRITE(2,55) AI,BI
        I=I+1
        GO TO 45
60      STOP
        END
```

Data

```
4     2     1
5.
9.
3.
-2.
1.
5.
10.
10.
4.
1.
2.
2.              2.
```

$$\frac{N(s)}{D(s)} = \frac{5s^3+9s^2+3s-2}{s^4+5s^3+10s^2+10s+4} = \frac{K_1}{s+2} + \frac{K_2}{s+2} + \frac{As+B}{s^2+2s+2}$$

(FIND)

Results

```
-.10000E+01
 .60000E+01
 .00000E+00  -.50000E+01
```

$K_1 = -1$

$K_2 = 6$

$A = 0 \quad B = -5$

Data

```
4     0     2
9.
32.15
241.31
125.5
1.
6.03
59.34
151.5
631.25
3.              25.
3.03            25.25
```

$$\frac{N(s)}{D(s)} = \frac{9s^3+32.15s^2+241.31s+125.5}{s^4+6.03s^3+59.34s^2+151.5s+631.25}$$

Results

```
.50000E+01   .20000E+01
.40000E+01   .30000E+01
```

$$= \frac{5s+2}{s^2+3s+25} + \frac{4s+3}{s^2+3.03s+25.25}$$

Data

```
2       2       0
2.
1.
1.
4.05
4.1
2.
2.05
```

$$G(s)=\frac{2s+1}{(s+2)(s+2.05)}$$

Results

```
-.60000E+02
 .62000E+02
```

PROBLEM: Draw the "Foster Form" of the control system by calc for K_1 and K_2



Data

```
2       2       0
2.
1.
1.
4.00001
4.00002
2.
2.00001
```

Impedance

$$Z(s)=\frac{2s+1}{(s+a)(s+b)}$$

$a=2 \quad b=2.00001$

Result

```
-.29959E+06
 .29960E+06
```

Problem: Produce a passive network (if possible!) represented by $Z(s)$ i.e.



where: $Z(s)=\frac{K_1}{s+a}+\frac{K_2}{s+b}$

```
2       2       0
2.
1.
1.
4.0000000001
4.0000000002
2.
2.0000000001


"FPNT" ERROR AT LOC 03152
```